

Paper 183-30

CUSTOMIZING ODS STATISTICAL GRAPHICS

Shi-Tao Yeh, GlaxoSmithKline, King of Prussia, PA

ABSTRACT

SAS® 9.1 includes an experimental feature, ODS Statistical Graphics (ODS Graphics for short), which introduces more than 30 procedures in four SAS products. This new feature enables the procedures to create statistical graphs automatically as part of ODS output.

With simple ODS Graphics statements, a procedure creates graphs that are commonly needed for data analysis. The default ODS Graphics provide informative graphs with important statistics printed alongside. Unfortunately, the default ODS Graphics are not suitable for the study reports, publications, or New Drug Application (NDA) submissions. A typical challenge is that the ODS Graphics are missing study specific titles and footnotes. To resolve this issue, the modification of default ODS Graphics is necessary.

ODS Graphics appearances are governed by the ODS Graphical Template Language (GTL). The modification of ODS Graphics can be made by three sources: 1) from the data set that is used for the procedure execution, 2) from changing the style template, and 3) from modification of the graph template.

This paper discusses several methods and tips to create customized ODS Graphics and how to modify system default templates or create stand alone template. The discussion focuses in four topics: 1) A macro facility for users to easily supply titles and footnotes, 2) Overlaying two ODS Graphics, 3) Modification of the ODS Graphics elements, and 4) Creating stand alone template. Some sample code and output are included in the paper for illustration purposes.

INTRODUCTION

The ancient Chinese proverb "a picture is worth a thousand words" describes the power of a picture. Graphics are pictures and are often a good means of communicating information. Statistical graphics represent statistical information and illustrate or compare trends. Good pictures offer clearly defined statistical results and elegantly represented information. They also offer sufficient detail for optimum clarity and eliminate confusion.

ODS Statistical Graphics is a new experimental feature in SAS 9.1. It provides commonly used statistical graphics, such as scatter plots, histograms, box plots, contour plots, and 3-D plots from several SAS products procedures automatically. The creation of ODS Graphics is as simple as one short step.

ODS Graphics can not produce just graphs without other ODS output tables and listings. In other words, ODS Graphics are part of ODS output and use at least one ODS destination to obtain graphics. ODS Graphics uses Java technology and is independent of SAS/GRAPH. ODS Graphics do not support the following statements: 1) any SAS/GRAPH statements, such as GOPTIONS, SYMBOL, PATTERN; 2) the GTITLE or GFOOTNOTE options available with the ODS destinations HTML, RTF, and MACKUP, and 3) the ODS USEGOPT statement.

The procedures supporting ODS Graphics from the SAS products of SAS/STAT®, Base SAS®, SAS/ETS® and SAS High-Performance Forecasting are listed below.

SAS Product	Procedures Supporting ODS GRAPHICS
Base SAS	CORR
SAS/STAT	ANOVA, CORRESP, GAM, GENMOD, GLM, KDE, LIFETEST, LOESS, LOGISTIC, MI, MIXED, PHREG, PRINCOMP, PRINQUAL, REG, ROBUSTREG
SAS/ETS	ARIMA, AUTOREG, ENTROPY, EXPAND, MODEL, SYSLIN, TIMESERIES, UCM, VARMAX, X12
SAS High-Performance Forecasting	HPF

Table 1. SAS Products and Procedures Supporting ODS Graphics

The syntax of ODS Graphics are described as follows:

```
ods graphics on</options>;
procedure specific ods graphics option statement;
ods graphics off;
```

ODS GRAPHICS ON</options> statement enables ODS Graphics with options sub-statement. ODS GRAPHICS OFF statement disables ODS Graphics. These two statements are a pair statement to open and close ODS Graphics request. In between, you can use the other ODS statements to produce ODS tables and listings. Each ODS graph has a graph name associated with it. Each procedure has procedure specific ODS Graphics option statement to process the ODS Graphics generation. Table 2 illustrates some procedure specific statements with sample code.

Product	Procedure	Procedure Specific ODS Graphics Statement	Sample Code
Base SAS	corr	proc	proc corr data=test plots;
SAS/STAT	lifetest	survival	survival plots=(s epb);
	phreg	assess	assess var=(x) ph/crpanel resample seed=19;
	logistic	model	model y=x z / influence iplots;
	kde	univar, bivar	bivar x y /plots=contour surface;
	loess	model	model y = x / smooth=0.0565 residual;
	reg	proc	proc reg data=test plots(unpack);
	mixed	model	model y = x / residual;
	corr	none	
	corresp	none	
	gam	none	
	genmod	assess	assess var=(x) /crpanel resample seed=19;
	glm	none	
	mi	mcmc	mcmc timeplot(mean(y1)) acfplot(mean(y1));
	princomp	none	
	prinqual	proc	proc prinqual data=test1 out=o1 n=2 replace mdpref ;
	robustreg	proc	proc robustreg data=test plot=reshistogram;

Table 2. ODS Graphics Procedure Specific Statement with Sample Code

You can see in Table 2 that there is no single, unique syntax to call ODS Graphics. Each procedure has different syntax. Some procedures use a new ODS Graphics statement to call ODS Graphics, others use PROC or MODEL statements to put ODS Graphics options.

With simple ODS Graphics statements shown above, a procedure creates graphs that are commonly needed for data analyses. It is a simple and powerful tool for a beginning step of data analysis. The default ODS Graphics provide informative graphs with important statistics printed alongside. Unfortunately, the default ODS Graphics are not suitable for the study reports, publications, or New Drug Application (NDA) submissions. A typical challenge is that the study specific titles and footnotes are missing in the default ODS Graphics or compliance with regulatory submission guidelines. To resolve this issue, the modification of default ODS Graphics is necessary.

ODS Graphics appearances are governed by the ODS Graphical Template Language (GTL). The modification of ODS Graphics can be made by three sources: 1) from the data set that is used for the procedure execution, 2) from changing the style template, and 3) from modification of the graph template.

This paper discusses several methods and tips to create customized ODS Graphics and how to modify system default templates. The discussion focuses in three topics: 1) A macro facility for users to easily supply titles and footnotes, 2) Overlaying two ODS Graphics together as one graph, and 3) Modification of the ODS Graphics elements. Some sample code and output are included in the paper for illustration purposes.

The ODS destinations that support ODS Graphics include the following:

Destination	Image Format Supported
HTML	GIF(default),JPEG,PNG
LATEX	PS(default), EPSI, GIF,JPEG,PNG
PCL	Not applicable
PDF	Not applicable
PS	Not applicable
RTF	Not applicable

Table 3. ODS Destination that Support ODS Graphics

ODS GRAPHICS FROM PROC LIFETEST

The following ODS GRAPHICS are provided from procedure LIFETEST.

ODS Graph Name	Plot Description	Statement	Method	PLOTS=(type)
Density	Density estimate	SURVIVAL	Life-table	PDF
Hazard	Hazard estimate	SURVIVAL	Life-table	HAZ
Stratum	4-panel of individual stratum displays; plots of survival curve, 95% HW Band and EP Band with pointwise limits in three panels and one panel of censored statistics	SURVIVAL	Life-table	STRATUM
EqualPrecision	Equal precision band overlaid with the estimate survival curve	SURVIVAL	Product-limit	EPB
HallWellner	Hall and Wellner band overlaid with the estimated survival curve	SURVIVAL	Product-limit	HWB
LogNegLogSurvival	Log of negative log of the estimated survival function	SURVIVAL	Product-limit	LLS
NegLogSurvival	Negative log of the estimated survival function	SURVIVAL	Product-limit	LS
Survival	Estimated survival curve	SURVIVAL	Product-limit	S
SurvivalCL	Pointwise confidence limits overlaid with estimated survival curve	SURVIVAL	Product-limit	CL
Stratum	4-panel of individual stratum displays; plots of estimated survival, estimated hazard, and estimated density with 95% limits in 3 panels and one panel of censored statistics	SURVIVAL	Life-table	STRATUM
Survival	Estimated survival curve	SURVIVAL	Life-table	S
SurvivalCL	Pointwise confidence limits overlaid with estimated survival curve	SURVIVAL	Life-table	CL

Table 4. ODS Graph Names and Description from PROC LIFETEST

ODS GRAPHICS SAMPLE OUTPUT FROM PROC LIFETEST

The following Sample Code 1 with the PROC LIFETEST procedure creates 3 default ODS Graphics (Survival, EqualPrecision and Hazard) which are shown in Figure 1

```
ods html path="c:\tip2"
         gpath="c:\tip2"
         file="a001ab.htm" ;
```

```
ods graphics on /imagefmt=jpeg imagename = "a001ab";
```

```
proc lifetest data=final ;
  time efsurvw * efsvflag(0);
  survival plots=(s epb);
  strata trtgrpt;
  id subjid;
run;
proc lifetest data=final method=lt;
  time efsurvw * efsvflag(0);
  survival plots=(haz);
  strata trtgrpt;
  id subjid;
run;
```





Figure 1. Default ODS Graphics from PROC LIFETEST Procedure

ODS GRAPHICS STYLE

ODS styles control the overall look of ODS output. ODS Graphics provides a style= option to produce different types of output appearance with the same content. There are four styles in SAS 9.1 for ODS Graphics output appearance: 1) default, 2) Journal, 3) analysis, and 4) statistical.






Figure 2. Sample ODS Graphics with Different Styles

CUSTOMIZING ODS GRAPHICS

In SAS 9.1, the ODS Graphics are controlled by a new graph definition language called ODS Graphics Template Language (GTL). ODS Graphics are part of procedure ODS output and are governed by the standard ODS statements. In addition, graph appearance, like ODS output tables, are controlled by the current ODS style.

The names of the templates a procedure uses are predefined, the users can not change a template's name or add new templates to a procedure's output. The default SAS supplied templates are located in SASHELP.TEMPLMST. You can change the default template and keep the same template name, but store the modified template to SASUSER path that is searched before the default path.

There are four sources for controlling ODS graph appearance: 1) from feed-in data set modification without changing GTL, 2) changing style template to control the general appearance of ODS Graphics, 3) modifying individual ODS graph template, and 4) creating non-procedure based stand alone template. Table 5 below demonstrates different sources for controlling ODS Graphics components.

Source for Control	ODS Graphics Component
Feed-in data set	X-axis label, y-axis label, legend value description
Style template	Colors, line styles, marker styles, font styles, graph sizes,
ODS graph template	Titles, footnotes, legend styles, line styles, colors styles, marker styles, font styles, borders and graph dimensions
Non-procedure based stand alone template	ODS Graphics template

Table 5. Source for Controlling ODS Graphics Components

These three sources and methods for modifying ODS Graphics are discussed in the following sections.

SOURCE I: DATA SET MODIFICATION WITHOUT CHANGING GTL

Figure 3 is a default hazard function estimates generated from Sample Code I.




Figure 3. Default ODS Graphics of Hazard Function Estimates

Two additional statements are added to the SAS data set; one is IF statement to hard code (N=xx) to value of variable trtgrpt, the other is LABEL statement to change the label for variable EFSURVW. The execution of the same Sample Code I produces Figure 4.

```
data final;
  set sb.mdata;
  if trtgrpt='Dose A' then trtgrpt='Dose A(N=67)';
  else if trtgrpt='Dose B' then trtgrpt='Dose B(N=20)';
  label efsurv='Survival Time (Weeks)';
```

There are several methods to insert the text of (N=xx) to the legend value description. Figure 4 illustrates the display (N=xx) as legend value description enhancement. The following sample code provides a different method for inserting the text of (N=xx) to the legend value description.

This method uses PROC SUMMARY to count the number of subjects in each treatment group then merges the output dataset from PROC SUMMARY to the target the dataset by treatment the group. You can create a new variable that contains (N=xx) as part of variable value.

```
data final;
  set final;
  count=1;
  run;
proc sort;by trtgrpt;
proc summary data=final;
  by trtgrpt;
  var count;
output out=allb sum=allb;
data final;
  merge final allb;
  by trtgrpt;
  length trt $15.;
  trt =left(trim(trtgrpt)) || ' (N=' || compress(allb) || ')';
```




Figure 4. Using SAS Data Set Modification to Customize Legend Value Description and Label of X Axis

SOURCE II: CUSTOMIZING STYLE TEMPLATE

ODS styles control the overall appearance of the ODS Graphics. A style definition provides formatting information for ODS output. This formatting information includes: 1) a list of font definition (each font defines a face, size, weight, and style), 2) a list of colors, 3) a list of line style, and 4) a list of marker style. The ODS Graphics elements are associated with these formatting information and summarized in Table 6.

Formatting Component	Graphics Element	Style Attribute
Graph Text	Titles, footnotes, labels, values, data	Foreground text color and font characteristics
Borders and Lines	Borders for the entire graph, legends, axis and grid lines, axis tick marks, minor, contours, reference lines, regression line, confidence lines and bands, prediction lines, standard error line or bar	Line color, line style and thickness, marker style and size, threshold, start neutral and end color for contours and gradient legends
Graph Size and Fill Colors	Graph size, background and walls, fills for prediction limits and confidence bands	Graph width, height, background and foreground color, filled area color, transparency, contrast color
Graphical Data	Data default (non-grouped data; areas, lines, markers), 1 st data group, 12 th data group, data and outlier data for the graph	Color of filled area, marker or line

Table 6. Style Formatting Info Associated with ODS Graphics Components and Elements

Sample Code II below uses PROC KDE with style=journal option to produce Figure 5.

```
ods html style=journal gpath="c:\tip2" file="b001aa.htm";
ods graphics on /imagefmt=jpeg imagename = "b001aa";
proc kde data=final ;
  bivar asat alat/plots = all;
run;
ods html close;
ods graphics off;
```




Figure 5. Contours Plot from PROC KDE

The partial listing of SAS supplied journal style template is listed as follows:

```
proc template;
  define style Styles.Journal;
    parent = styles.default;

    replace GraphFonts /
      'GraphDataFont' = ("Arial",8pt)
      'GraphValueFont' = ("Arial",10pt)
      'GraphLabelFont' = ("Arial",12pt)
      'GraphFootnoteFont' = ("Arial",12pt)
      'GraphTitleFont' = ("Arial",14pt);
    replace GraphColors /
      'gheader' = colors('docbg')
      'gconramp3cend' = cxF0F0F0
      'gconramp3cneutral' = cxA7A7A7
      'gconramp3cstart' = cx5F5F5F
      'gramp3cend' = cxF0F0F0
      'gramp3cneutral' = cxA7A7A7
      'gramp3cstart' = cx5F5F5F
      'gconramp2cend' = cx5F5F5F
      'gconramp2cstart' = cxF0F0F0
      'gramp2cend' = cxF0F0F0
      'gramp2cstart' = cx5F5F5F

      'gdata1' = CXBFBFBF;
      .
      .
      .

    replace GraphData1 /
      linestyle = 1
      contrastcolor = GraphColors('gdata1')
      foreground = GraphColors('gdata1');
      .
      .
      .
```



```

replace GraphData12 /
  linestyle = 39
  contrastcolor = GraphColors('gcdatal2')
  foreground = GraphColors('gdata12');
end;
run;

```

The modification of the style template statements are graph font and graph color as follows:

```

'GraphValueFont' = ("Arial",10pt,Bold)
'GraphLabelFont' = ("Arial",12pt,Bold)
'GraphTitleFont' = ("Arial",15pt,Bold)

'gconramp3cend' = cxFF0055
'gconramp3cneutral' = cxA7A7A7
'gconramp3cstart' = cx6666FF
'gramp3cend' = cxFF0055
'gramp3cneutral' = cxA7A7A7
'gramp3cstart' = cx6666FF

```

The customized output is shown in Figure 6.




Figure 6. Modification of Style Template for Customizing Contours Plot from PROC KDE

SOURCE III: CUSTOMIZING ODS GRAPH TEMPLATE

In SAS 9.1, each ODS Graphics is predefined an ODS graph name and a graphical template. You can not change a template's name or add new template to a procedure's output. The items you can not modify are: 1) template name, 2) the names of any DYNAMIC variables, and 3) column names that appear in the template. You can modify existing template contents such as graphical element attributes and stored the modified template in SASUSER path. Figure 7 shows SAS supplied ODS Graphics templates for procedure LIFETEST. Table 7 lists ODS Graphics name with associated template.




Figure 7. ODS Graphics Templates in Procedure LIFETEST

ODS Graph Name	Template Name	Method	PLOTS=(type)
Density	statgraph Stat.Lifetest.Graphics.Density;	Life-table	PDF
Hazard	statgraph Stat.Lifetest.Graphics.Hazard	Life-table	HAZ
Stratum	statgraph Stat.Lifetest.Graphics.LifeTablePanel	Life-table	STRATUM
EqualPrecision	statgraph Stat.Lifetest.Graphics.EqualPrecision;	Product-limit	EPB
HallWellner	statgraph Stat.Lifetest.Graphics.HallWellner	Product-limit	HWB
LogNegLogSurvival	statgraph Stat.Lifetest.Graphics.LogNegLogSurvival;	Product-limit	LLS
NegLogSurvival	statgraph Stat.Lifetest.Graphics.NegLogSurvival;	Product-limit	LS
Survival	statgraph Stat.Lifetest.Graphics.ProductLimit	Product-limit	S
SurvivalCL	statgraph Stat.Lifetest.Graphics.ProductLimitCL;	Product-limit	CL
Stratum	statgraph Stat.Lifetest.Graphics.ProductLimitPanel;	Product-limit	STRATUM
Survival	statgraph Stat.Lifetest.Graphics.LifeTable	Life-table	S
SurvivalCL	statgraph Stat.Lifetest.Graphics.LifeTableCL;	Life-table	CL

Table 7. Graph Names with Associated ODS Graphics Template Names in Procedure LIFETEST

Table 7 lists ODS Graphics name with associated template. The ODS Graphics template code below is SAS supplied Hazard template in procedure LIFETEST.

```

proc template;
  define statgraph Stat.Lifetest.Graphics.Hazard;
    dynamic NStrata;
    layout Gridded;
      if (NSTRATA=1)
        layout gridded / padbottom=5;
        EntryTitle "Hazard Function Estimate";
      endlayout;
    layout OVERLAY;
      seriesplot y=HAZARD x=MIDPOINT / linecolor=
        StatGraphFitLine:contrastcolor markercolor=
        StatGraphFitLine:contrastcolor index=STRATUMNUM markers=on name
="Hazard";
      EndLayout;
    else
      layout gridded / padbottom=5;
      EntryTitle "Hazard Function Estimates";
      endlayout;
    layout OVERLAY;
      seriesplot y=HAZARD x=MIDPOINT / group=STRATUM index=STRATUMNUM
        markers=on name="Hazard";
      DiscreteLegend "Hazard" / hAlign=right vAlign=top across=1 border=
true;
      EndLayout;
    endif;
  EndLayout;
end;
run;

```

The Hazard Function graph is modified in the following items: 1) three title lines instead of only one with user's provided titles, 2) change the position of legend from right top corner to left bottom corner, 3) a footnote is added, 4) line colors are added, and 5) thicker lines are selected.

The following sample code is the modified template. The modified code is highlighted. The output is shown in Figure 8.

```

proc template;
  define statgraph Stat.Lifetest.Graphics.Hazard / store = SASUSER.TEMPLAT;
    dynamic NStrata;

    layout gridded / padbottom=5 padleft=0;
    EntryTitle "Protocol: XYZ 99999 Study 12345" / hAlign=left fontsize=10pt;
    EntryTitle "Population: Intent-to-Treat" / hAlign=left fontsize=10pt;
    EntryTitle "Figure 12 Hazard Function Estimate"/nAlign=center fontsize=15pt;

    if (NSTRATA=1)
      layout OVERLAY;
      seriesplot y=HAZARD x=MIDPOINT / linecolor=
        StatGraphFitLine:contrastcolor markercolor=
        StatGraphFitLine:contrastcolor index=STRATUMNUM markers=on name ="Hazard";
      EndLayout;
    else
      layout lattice / rows=1 columns=1 border=false;
      sidebar / align=bottom;
      endsidebar;
      layout OVERLAY;
      seriesplot y=HAZARD x=MIDPOINT / group=STRATUM index=STRATUMNUM
        markers=on name="Hazard";
      DiscreteLegend "Hazard" / hAlign=left vAlign=bottom across=1 border=false;
      EndLayout;
    endif;
  entry "USERID:c:\tip2\lab.sas &sysdate &sysstime" / hAlign=left fontsize=10pt;

```

```

EndLayout ;
end;
run;

```




Figure 8. Sample ODS Graphics with Customization

OVERLAYING MULTIPLE ODS GRAPHICS

Figure 1 demonstrates 3 default ODS Graphics from PROC LFTEST. A desired output is to overlay Survival graph and Confidence Bands graph together. It can be achieved by inserting Confidence Bands plots statement to Stat.Lifetest.Graphic.ProductLimit template. The bands statements are as follows:

```

Band yLimitUpper=HW_UCL yLimitLower=HW_LCL x=TIME / group=
STRATUM index=STRATUMNUM modelname="Survival" fill=true lines=
true datatransparency=0.8;

```

The output is shown in Figure 9.




Figure 9. Overlaying ODS Graphics

CUSTOMIZING ODS GRAPHICS MADE EASY – A SAS MACRO APPROACH

```

%macro haz(hkt1=,hkt2=,t1=,t2=,f1=,f2=,hkf1=);
proc template;

```

```

define statgraph Stat.Lifetest.Graphics.Hazard;
  dynamic NStrata;
  layout Gridded;
    layout gridded / padbottom=5 padleft=0;
      Entry "Protocol: &hkt1"
        /hAlign=left fontsize=10pt;
      Entry "Population: &hkt2"
        /hAlign=left fontsize=10pt;
      EntryTitle "&t1 " /fontsize=15pt;
      EntryTitle "&t2" /fontsize=15pt;

  if (NSTRATA=1)
    layout gridded / padbottom=5 ;

  endlayout;
  layout OVERLAY;
    seriesplot y=HAZARD x=MIDPOINT / linecolor=
      StatGraphFitLine:contrastcolor markercolor=
      StatGraphFitLine:contrastcolor index=STRATUMNUM markers=on name
      ="Hazard" linethickness=2px;
  EndLayout;
else
  layout gridded / padbottom=5 ;

  endlayout;
  layout lattice / rows=1 columns=1 border=false;

  layout OVERLAY;
    seriesplot y=HAZARD x=MIDPOINT / group=STRATUM index=STRATUMNUM
      markers=on name="Hazard" linethickness=2pt;
    DiscreteLegend "Hazard" / hAlign=left vAlign=bottom across=1 border=
      true;
  EndLayout;
  endif;
EndLayout;
  EntryFootnote "&f1" /hAlign=left fontsize=12pt;
  EntryFootnote "&f2" /hAlign=left fontsize=12pt;
  EntryFootnote "USERID:&hkf1 &sysdate &sysstime" /hAlign=left fontsize=10pt;

  endlayout;
end;
run;
%mend;
data final;
set sb.mdata;
run;
%let t1=Figure 12 Hazard Function Estimates;
%let hkt1=XYZ 99999 Study 12345;
%let hkt2=Intent-to-Treat;
%let hkf1=c:\tip2\a001ab.sas;
%global t1 hkt1 hkt2 hkf1;
ods html style=journal gpath="c:\tip2" file="a001ab.htm";
ods graphics on /imagefmt=jpeg imagename = "a001ab";
proc lifetest data=final method=lt;
time efsurvw * efsvflag(0);
survival plots=(haz);
strata trtgrpt;
id subjid;
run;
ods html close;
ods graphics off;

```

CREATING YOUR OWN ODS GRAPHICS TEMPLATE

If a procedure does not produce a graph you like, you can create your own stand alone graphics template to produce the desired graph. Table 8 lists the detailed steps to accomplish this task.

Step	Task
I	Using procedure(s) to create output data sets with analyzed results.
II	Defining a graphics template unassociated with any procedure.
III	Compiling the template
IV	Using data step to read in the output data sets
V	Using FILE and PUT statements to feed the output data set to your graphics template.

Table 8. Using Stand Alone Template to Create ODS Graphics

Sample Code: Step I

```
libname sb 'c:\tip2';

data final;
set sb.mdata;
run;
proc lifetest data=final method=lt outsurv=outs;
time efsurvw * efsvflag(0);
survival stderr out=hzstderr;
strata trtgrpt;
id subjid;
run;
data hzstderr;
set hzstderr;
haz_su=hazard + haz_stderr;
haz_sl=hazard - haz_stderr;
keep trtgrpt midpoint hazard haz_su haz_sl stratum stratumnum;
label midpoint='Survival(Weeks)';
```

Sample Code: Step II

```
%macro haz(hkt1=,hkt2=,t1=,t2=,f1=,f2=,hkfl=);
proc template;
define statgraph mygraphs.hazardse;
layout Gridded;
Entry "Protocol: &hkt1"/hAlign=left fontsize=10pt;
Entry "Population: &hkt2" /hAlign=left fontsize=10pt;
EntryTitle "&t1 " /fontsize=15pt;
EntryTitle "&t2" /fontsize=15pt;
layout gridded / padbottom=5 padleft=0;
layout lattice / rows=1 columns=1 border=false;
layout OVERLAY;
Band yLimitUpper=haz_su yLimitLower=haz_sl x=MIDPOINT / group=STRATUM
index=STRATUMNUM fill=true lines=true datatransparency=0.8;
seriesplot y=HAZARD x=MIDPOINT / group=trtgrpt index=STRATUMNUM
markers=on name="Hazard" linethickness=2pt;
DiscreteLegend "Hazard" / hAlign=left vAlign=top across=1 border=
true;
Endlayout;
Endlayout;
EntryFootnote "&f1"/hAlign=left fontsize=12pt;
EntryFootnote "&f2"/hAlign=left fontsize=12pt;
EntryFootnote "USERID:&hkfl &sysdate &sysstime"/hAlign=left fontsize=10pt;
```

```

        endlayout;
    end;
run;
%mend;

```

Sample Code: Step III

```

%let t1=Figure 12 Hazard Function Estimates with Standard Error Bands;
%let hkt1=XYZ 99999 Study 12345;
%let hkt2=Intent-to-Treat;
%let hkf1=c:\sugi30\a01hz.sas;
%global t1 hkt1 hkt2 hkf1;
%haz(hkt1=&hkt1, hkt2=&hkt2,t1=&t1, hkf1=&hkf1);

```

Sample Code: Steps VI and V

```

ods html style=journal path="c:\sugi30" gpath="c:\sugi30" file="a01hz.htm";
ods graphics on /imagefmt=jpeg imagename = "a01hz";
data _null_;
set hzstderr;
file print ods=(template="mygraphs.hazardse" );
put _ods_;run;
ods html close;
ods graphics off;

```




Figure 10. ODS Graphics Output Produced from Stand Alone Template

Several non-procedure based stand alone templates are developed for survival analyses. The descriptions of these templates are listed in Table 9. The output data are from procedures of LIFETEST and PHREG. The graphics output are shown in Figure 11.

Data Used	Template Name	Graphics Description
PROC LIFETEST / method=lt	mygraphs.ltpan	<ul style="list-style-type: none"> • Produces a 4-panel display; • Plots of estimated survival with 95 % confidence bands, estimated hazard functions with +/- SE estimated probability density functions with +/- SE, in 3 panels. • A panel of censored summary and test of equality over strata statistics.

PROC LIFETEST / method=pl	mygraphs.plpan	<ul style="list-style-type: none"> • Produces a 4-panel display; • Plots of estimated survival with 95 % confidence bands, negative log of estimated survival functions with +/- SE and log of negative log of estimated survival functions with +/- SE, in 3 panels. • A panel of censored summary and test of equality over strata statistics.
PROC LIFETEST / method=lt	mygraphs.ltpan6	<ul style="list-style-type: none"> • Produces a 4-panel display; • Plots of estimated survival with 95 % confidence bands, estimated hazard functions with +/- SE estimated probability density functions with +/- SE, and conditional probability of failure with +/- SE in 4 panels.
PHREG / method=pl	mygraphs.phrega	<ul style="list-style-type: none"> • Produces a 4-panel display; • Plots of estimated survival functions, negative log of estimated survival, in 4 panels.

Table 9. Stand Alone Template Description




Figure 11. Sample Output from Stand Alone Templates

CONCLUSIONS

Graphics output is an effective means to convey the information. The ability to easily produce a default ODS Graphics from a procedure can be achieved by SAS V9.1 experimental feature of ODS Graphics. But the modification of default ODS Graphics is necessary in many situations.

This paper provides the following methods to modify default ODS Graphics:

- * Use the fed-in data set to modify x-axis label, y-axis label, and legend value description without any change of Template.
- * Change the SAS-supplied style template for general graphical appearance.
- * Modify the procedure-based graph template to fit your needs.
- * Create stand alone template to produce the desired graphics
- * Combine above 4 methods together to achieve the greatest results.

REFERENCES

- [1] Cartier, J.: *It's All in the Presentation*, Proceedings of Annual Conference of the SAS Users Group International, SUGI 28, Paper 147-28, April 2003
- [2] CDISC website: <http://www.cdisc.org/>
- [3] Delaney, K.: *Multiple Graphs on One Page, the easy way (PDF) and the hard way (RTF)*, Proceedings of Annual Conference of the SAS Users Group International, SUGI 28, Paper 94-28, April 2003
- [4] Johnston, G and Y. So: *Let the Data Speak: New Regression Diagnostics Based on Cumulative Residuals*, Proceedings of Annual Conference of the SAS Users Group International, SUGI 28, Paper 255-28, April 2003
- [5] Mitchell, R.M.: *Thumbs Up For ODS Graphics, But Don't Throw Out All Your SAS/GRAPH Program*, Proceedings of Annual Conference of the SAS Users Group International, SUGI 29, Paper 083-29, May 2004
- [6] Robert N. Rodriguez: *An Introduction to ODS for Statistical Graphics in SAS 9.1*, Proceedings of Annual Conference of the SAS Users Group International, SUGI 29 Paper 204-29, May 2004
- [7] <http://support.sas.com/rnd/base/topics/statgraph/v91StatGraphStyles.htm>
- [8] SAS Institute Inc.: *SAS 9.1 Output Delivery System: User's Guide*, SAS Institute Inc., Cary NC, USA. Jan. 2004
- [9] SAS Institute Inc.: *The Analyst Application, Second Edition*, SAS Institute Inc., Cary NC, USA. 2003
- [10] SAS Institute Inc.: *SAS/STAT 9.1 User's Guide*, SAS Institute Inc., Cary NC, USA. Jan. 2004
- [11] SAS Institute Inc.: *The Complete Guide to the SAS Output Delivery System, Version 8*, SAS Institute Inc., Cary NC, USA. 1999
- [12] Shannon, D.: *To ODS RTF and Beyond*, Proceedings of Annual Conference of the SAS Users Group International, SUGI 27, Paper 1-27, April 2002
- [13] Watt, P.: *Multiple-Plot Displays: Simplified with Macros*, SAS Publishing, Books by Users press. 2004
- [14] Yeh, S.: *Tips to Enhance Your SAS Statistical Graphics Output*, Proceedings of Annual Conference of the Pharmaceutical Industry SAS Users Group, PharmaSUG 2003, May 2003, pp 343-348
- [15] Zhou, J.: *An Enhancement of Survival Graph* Proceedings of Annual Conference of the Pharmaceutical Industry SAS Users Group, PharmaSUG 2003, May 2003, pp 403-408

TRADEMARKS

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are registered trademarks or trademarks of their respective companies.

AUTHOR CONTACT INFORMATION

Shi-Tao Yeh, Ph. D.
 (610)787-3856 (W)
 E-mail: shi-tao_yeh-1@gsk.com